Appendix 2

JAGS language description of Florida Sandhill Crane BBS trend model

# Data input as list include:
# count = Sandhill crane counts per route as matrix of route ID (rows) x year (columns).
# nrts = number of routes.
# styrnum = start year as an integer.
# enyrnum = end year as an integer.
# medyr = median of yr vector.
# div = vector of integers representing each route's PDSI division.
# PDSI = winter PDSI values as matrix of Florida PDSI divisions (rows) x year (columns).
# PDSI_lag = 1-year lagged winter PDSI values (i.e., from year preceding survey) as matrix of Florida PDSI divisions (rows) x year (columns).
# deltaland = percent land cover change for each route for one of six cover types (grassland, wetland, woodland, scrub, urban or other).

model{
  #Overdispersed poisson model:
  for (i in 1:nrts){
    for (t in styrnum:enyrnum){
      log(lambda[i,t]) <- S[i]+beta[i]*(t-medyr)+gamma*PDSI[div[i],t]+gammalag*PDSI_lag[div[i],t]+eps[i,t]
      count[i,t] ~ dpois(lambda[i,t])
      eps[i,t] ~ dnorm(0, tau.epsilon)
    }
  }
  #beta[i] sub model:
  for(i in 1:nrts){
    beta[i] <- b[i] #for 1996- and 1985-2016 models without land cover change effect
#beta[i] <- phi*deltaland[i] + b[i] #for 1985-2016 models with landcover change effect

b[i] ~ dnorm(B1,tau.b)

} #Priors:

for (i in 1:nrts){
    S[i] ~ dunif(-4,4)
}

tau.b ~ dgamma(0.001,0.001)

phi ~ dnorm(0,0.01)

B1 ~ dnorm(0,0.01)

gamma ~ dnorm(0,0.01)

gammalag ~ dnorm(0,0.001)

tau.epsilon ~dgamma(0.001,0.001)

} #end model